


Techniki mikroprocesorowe, laboratorium 2
Autorzy: Konrad Sobolowski, Bartłomiej Boczek

Treść zadania

Należy zaprojektować i zrealizować licznik 4-bitowy – ładowanie, zliczanie w górę – kod Graya

SCHEMAT POŁĄCZENIA ZADANIA Z LABORATORIUM NR 2



Komentarz- jako że na płytce laboratoryjnej, dostępne są jedynie wejścia OEC, OEB i CLKa, to na schemacie zaznaczyliśmy tylko te wejścia, pamiętając, że aby układ działał LE dla 573 musi być wysoki, i OE dla 574 niski (ponieważ, OE jest aktywowany sygnałem niskim)

```

1  # File lab2.asm
2  org 1800h
3  ;program do ładowania i zwiększania licznika w kodzie Gray'a.
4  ;Rejestr D - licznik w binary; F - licznik w Gray;
5  ;E - służy do rozpoznawania czy inkrementujemy podczas przerwania wywołanego przez przyciski 1 (wczytujący z hexów na lampki);
6  ;H i B - czekanie podczas drgań styków
7  ;C - adres lampek;
8  START:
9      LD SP,1900h
10     jr INITIALIZATION
11
12     ds 0x1838-$$,0
13
14     INT: ;przerwanie: wciśnięty klawisz nr 2, użytkownik wczytuje cyfrę z hex
15         EX AF, AF'
16         IN A,(01) ;wczytujemy wejścia do A
17         LD D,A ;początek procedury zamiany binary -> gray
18         SRL A ;przesuwa bity A o jeden w prawo bez przeniesienia
19         AND 7 ;upewniamy się że pierwszy bit jest zerem
20         XOR D ;xor A z D, zapis do A, teraz już powinien być w A w gray
21         AND 15
22         LD L,A ;wynik w gray przechowujemy w L
23         OUT (01), A ;wyświetlanie na lampkach stanu 01
24         IN A,(01) ;
25         BIT 4, A ;sprawdzamy, czy klawisz inkrementacji jest wciśnięty (rejesrt 573->wejście numer 4, czyli 5 z kolei) BIT ->
        testuj bit
26         LD E, 0 ;zerujemy flagę ;LD nie wpływa na zmianę znacznika flag
27         JR NZ, KONIEC_INT ;jeśli instrukcja BIT dała 1-guzik nie jest wciśnięty (w rejestrze A na 4 bicie jest 1) wykonuje się skok
        warunkowy
28         LD E, 1 ;jeżeli jump (JR) się nie wykona to ustawiamy flagę na 1
29
30     KONIEC_INT: ;wychodzimy z INT
31         EX AF, AF' ;wymień ze sobą pary rejestrów
32         EI ;włączamy przerwania
33         RETI ;powrót ze śladu na stosie
34
35     ds 0x1860-$$,0
36
37     INITIALIZATION:
38         IM 1 ;tryb 1 przerwań
39         EI ;przerwania włączone
40         IN A,(01) ;ładujemy to co jest w hexie na lampki
41
42         LD D,A ;zamiana w gray
43         SRL A
44         AND 7
45         XOR D
46         AND 15
47         LD L,A ;koniec zamiany, w L jest liczba z pokręteł hex
48         LD E, 0 ;zerujemy rejestr E, inicjalizujemy flagę E na 0, czyli nie jesteśmy w przerwaniu
49
50     GLOWNA_PETLA:
51         LD A,L
52         OUT (01), A ;wyswietla na lampkach zawartosc L
53         IN A,(01) ;ładujemy do A stan przycisków
54         BIT 4, A ;sprawdzamy, czy został wciśnięty przycisk od inkremetnacji
55         JR NZ, GLOWNA_PETLA ;jeśli nie, powtarzamy pętlę
56         LD H, 250 ;licznik prób
57         LD B, 200 ;licznik udanych prób
58
59     DRGANIA_STYKOW:
60         DEC H ;zmniejsza H i ustawia znacznik Z na 1 jeśli wynikiem dekrementacji jest 0
61         JR Z, GLOWNA_PETLA ;jeśli H = 0 - klawisz nie wciśnięty i wracamy do głównej pętli
62         IN A,(01) ;wczytujemy guzik
63         BIT 4, A ;sprawdzenie czy guzik 2(od inkrementacji) jest wciśnięty ustawiamy znacznik Z
64         JR NZ, DRGANIA_STYKOW ;jeśli not Z switch w tym momencie niewciśnięty, DRGANIA_STYKOW od początku
65         DEC B ;klawisz wciśnięty, dekrementuje B i ustawia znacznik Z
66         JR Z, INKREMENTUJ_LICZNIK ;jeśli B = 0 -> klawis NAPRAWDĘ wciśnięty i idź do etykiety INKREMENTUJ_LICZNIK
67         LD A,40
68     CZEKAJ: ;pętla w pętli - okazało się, że 200 zliczeń na 250 to stanowczo za mało
69         DEC A
70         JR NZ, CZEKAJ
71         JR DRGANIA_STYKOW ;B jeszcze różne od zera, od początku ta etykieta
72
73     INKREMENTUJ_LICZNIK:
74         DI
75         BIT 0, E ;sprawdzam, czy wciśnięty podczas przerwania
76         LD E, 0 ;zeruje flagę
77         JR NZ, DDD ;jeśli flaga = 1, nie inkrementuj i wyjdź do głównej pętli
78         INC D
79         LD A,D ;początek zamiany w gray : kopiuj D do A
80         SRL A ;przesuwa bity A o jeden w prawo bez przeniesienia
81         AND 7
82         XOR D
83         AND 15
84         LD L,A
85         OUT (01), A ;wyswietla na lampkach zawartosc L
86         LD A,D
87         CP 16 ;gdy D==16, Z=1 ładuj na nowo sprawdzamy czy w A i 16 (CP) jest to samo
88         JR NZ, DDD ;gdy nie ma potrzeby ładowania wróć do głównej pętli
89         LD D,0
90         LD A,0
91         LD L,A
92         OUT (01), A
93
94     DDD:
95         EI ;aby po wciśnięciu klawisza nie zliaczalo
96         IN A,(01) ;pobranie tego co jest w latchu
97         BIT 4,A
98         JR Z, DDD
99         JP GLOWNA_PETLA ;wracamy do głównej pętli

```